' LangageGraphique, version 2.0

' Copyright (C) 2006 André Boileau

' Macros créées par André Boileau (août 2003) (dernière revision: GRMS mai 2010)

' Référence: http://www.math.uqam.ca/_boileau/LangageGraphique.html

' Ce programme est libre, vous pouvez le redistribuer et/ou le modifier

' selon les termes de la Licence Publique Générale GNU publiée par

' la Free Software Foundation (version 2 ou bien toute autre version ultérieure choisie par vous).

' Ce programme est distribué car potentiellement utile, mais SANS AUCUNE GARANTIE,

' ni explicite ni implicite, y compris les garanties de commercialisation ou d'adaptation

' dans un but spécifique. Reportez-vous à la Licence Publique Générale GNU pour plus de détails.

Public Const MiseAuPointLG As Boolean = False

Public Const pi As Double = 3.1415926535898

Dim TortueX As Double, TortueY As Double, TortueAngle As Double, TortueCrayon As Boolean

Dim CentreX As Double, CentreY As Double

Dim TortueR As Integer, TortueG As Integer, TortueB As Integer, TortueEpais As Integer

Dim TortueRemplis As Boolean, RemplissageR As Integer, RemplissageG As Integer, RemplissageB As Integer

Dim Points(10000, 1) As Double, NbPoints As Integer

Dim ProFacteur As Double

Dim FonctionsXmin As Double, FonctionsXmax As Double, FonctionsYmin As Double, FonctionsYmax As Double

Dim FenetreXmin As Double, FenetreXmax As Double, FenetreYmin As Double, FenetreYmax As Double

Dim FonctionsMX As Double, FonctionsBX As Double, FonctionsMY As Double, FonctionsBY As Double

Dim FonctionsMXE As Double, FonctionsBXE As Double, FonctionsMYE As Double, FonctionsBYE As Double

Dim FonctionsPas As Double, FonctionsPointsRelies As Boolean

Dim NbTotalImagesFilm As Integer

Dim NomFenetre As String

Dim NomFenetreFilm As String

Public ModeDemo As Boolean

Sub Init()

    CentreX = 305

    CentreY = 390

    TortueX = 0

    TortueY = 0

    TortueAngle = 0

    TortueCrayon = True

    TortueR = 0

    TortueG = 0

    TortueB = 0

    TortueEpais = 1

    TortueRemplis = False

    RemplissageR = 0

    RemplissageG = 0

    RemplissageB = 0

    NbPoints = 0

    ProFacteur = 0.5

End Sub

Sub EffacerTableau()

    Selection.WholeStory

    Selection.TypeBackspace

End Sub

Sub VE()

    Init

    ActiveDocument.Shapes.AddLine(0, 0, 0, 0).Select

    ActiveDocument.Shapes.SelectAll

    Selection.ShapeRange.Delete

End Sub

Sub ViderEcran()

    VE

End Sub

Sub Segment(x1, y1, x2, y2)   ' Accepte les nombres décimaux

    If TortueRemplis Then

        If NbPoints = 0 Then

            Points(0, 0) = x1 + CentreX

            Points(0, 1) = CentreY - y1

            NbPoints = 1

        End If

        Points(NbPoints, 0) = x2 + CentreX

        Points(NbPoints, 1) = CentreY - y2

        NbPoints = NbPoints + 1

        If MiseAuPointLG Then

            ActiveDocument.Shapes.AddLine(x1 + CentreX, CentreY - y1, x2 + CentreX, CentreY - y2).Select

            Selection.ShapeRange.Line.ForeColor.RGB = RGB(TortueR, TortueG, TortueB)

            Selection.ShapeRange.Line.Weight = TortueEpais

        End If

    Else

        ActiveDocument.Shapes.AddLine(x1 + CentreX, CentreY - y1, x2 + CentreX, CentreY - y2).Select

        Selection.ShapeRange.Line.ForeColor.RGB = RGB(TortueR, TortueG, TortueB)

        Selection.ShapeRange.Line.Weight = TortueEpais

    End If

End Sub

Sub Poly()

    Dim i As Integer

    On Error GoTo Erreur

    If NbPoints < 2 Then Exit Sub

    With ActiveDocument.Shapes.BuildFreeform(msoEditingAuto, Points(0, 0), Points(0, 1))

        For i = 1 To NbPoints - 1

            .AddNodes msoSegmentLine, msoEditingAuto, Points(i, 0), Points(i, 1)

        Next i

        .ConvertToShape.Select

    End With

    Selection.ShapeRange.Line.Visible = TortueCrayon  ' msoTrue ou msoFalse

    Selection.ShapeRange.Line.Weight = TortueEpais

    Selection.ShapeRange.Line.ForeColor.RGB = RGB(TortueR, TortueG, TortueB)

    If TortueRemplis Then Selection.ShapeRange.Fill.ForeColor.RGB = RGB(RemplissageR, RemplissageG, RemplissageB)

    Exit Sub

Erreur: TortueRemplis = False

    PolyDegenere

End Sub

Sub PolyDegenere()

    Dim i As Integer

    If NbPoints < 2 Then Exit Sub

    For i = 0 To NbPoints - 2

        ActiveDocument.Shapes.AddLine(Points(i, 0), Points(i, 1), Points(i + 1, 0), Points(i + 1, 1)).Select

        Selection.ShapeRange.Line.ForeColor.RGB = RGB(TortueR, TortueG, TortueB)

        Selection.ShapeRange.Line.Weight = TortueEpais

    Next i

End Sub

Sub Ellipse(x1, y1, x2, y2, Remplir, Angle)

    ActiveDocument.Shapes.AddShape(msoShapeOval, x1 + CentreX, CentreY - y2, x2 - x1, y2 - y1).Select

    Selection.ShapeRange.Line.Weight = TortueEpais

    Selection.ShapeRange.Line.ForeColor.RGB = RGB(TortueR, TortueG, TortueB)

    Selection.ShapeRange.IncrementRotation (Angle Mod 360)

    If Remplir Then

        Selection.ShapeRange.Fill.ForeColor.RGB = RGB(RemplissageR, RemplissageG, RemplissageB)

        Selection.ShapeRange.Fill.Visible = msoTrue

        Selection.ShapeRange.Fill.Solid

    Else

        Selection.ShapeRange.Fill.Visible = msoFalse

    End If

End Sub

Sub Cercle(x, y, r)

    Ellipse x - r, y - r, x + r, y + r, False, 0

End Sub

Sub disque(x, y, r)

    Ellipse x - r, y - r, x + r, y + r, True, 0

End Sub

Sub Arc(x, y, r, depart, arrivee)

    ActiveDocument.Shapes.AddShape(msoShapeArc, x + CentreX, CentreY - r - y, r, r).Select

    Selection.ShapeRange.Adjustments.Item(1) = arrivee

    Selection.ShapeRange.Adjustments.Item(2) = depart

    Selection.ShapeRange.Line.Weight = TortueEpais

    Selection.ShapeRange.Line.ForeColor.RGB = RGB(TortueR, TortueG, TortueB)

End Sub

Sub Av(d)

    Dim x As Double, y As Double

    x = TortueX + d * Sin(TortueAngle * pi / 180)

    y = TortueY + d * Cos(TortueAngle * pi / 180)

    If TortueCrayon Or TortueRemplis Then Segment TortueX, TortueY, x, y

    TortueX = x

    TortueY = y

End Sub

Sub Avance(d)

    Av d

End Sub

Sub Fd(d)

    Av d

End Sub

Sub Dr(a)

    TortueAngle = TortueAngle + a

    If TortueAngle > 360 Then TortueAngle = TortueAngle - 360

    If TortueAngle < -360 Then TortueAngle = TortueAngle + 360

End Sub

Sub Droite(a)

    Dr a

End Sub

Sub Rt(a)

    Dr a

End Sub

Sub FPos(x, y)

    If TortueCrayon Or TortueRemplis Then Segment TortueX, TortueY, x, y

    TortueX = x

    TortueY = y

End Sub

Sub FixePosition(x, y)

    FPos x, y

End Sub

Function PosX()

    PosX = TortueX

End Function

Function PosY()

    PosY = TortueY

End Function

Sub FCap(a)

    TortueAngle = a

End Sub

Sub FixeCap(a)

    FCap a

End Sub

Function Cap()

    Cap = TortueAngle

End Function

Function CapVers(x, y)

    x = x - TortueX

    y = y - TortueY

    If x = 0 Then

        CapVers = Sgn(y) * 90 - 90

    Else

        CapVers = 90 * Sgn(x) - Atn(y / x) * 180 / pi

    End If

End Function

Sub LC()

    TortueCrayon = False

End Sub

Sub LeveCrayon()

    LC

End Sub

Sub BC()

    TortueCrayon = True

End Sub

Sub BaisseCrayon()

    BC

End Sub

Sub CC(r, G, b)

    TortueR = r

    TortueG = G

    TortueB = b

End Sub

Sub CouleurCrayon(r, G, b)

    CC r, G, b

End Sub

Sub TC(d)

    TortueEpais = d

End Sub

Sub TailleCrayon(d)

    TC d

End Sub

Sub DebutRemplir()

    If TortueRemplis Then Exit Sub

    TortueRemplis = True

    NbPoints = 0

End Sub

Sub FinRemplir()

    Poly

    TortueRemplis = False

End Sub

Sub CR(r, G, b)

    RemplissageR = r

    RemplissageG = G

    RemplissageB = b

End Sub

Sub CouleurRemplissage(r, G, b)

    CR r, G, b

End Sub

Sub Dessous()

    On Error Resume Next

    Selection.ShapeRange.ZOrder msoSendToBack

    On Error GoTo 0

End Sub

Sub Dessus()

    On Error Resume Next

    Selection.ShapeRange.ZOrder msoBringToFront

    On Error GoTo 0

End Sub

Sub Texte(chaine, x, y, Police, taille, Gras, Italique)

    x = CentreX + x

    y = CentreY - y

    ActiveDocument.Shapes.AddTextbox(msoTextOrientationHorizontal, x, y, _

        Int(Len(chaine) * taille * 0.7), Int(taille * 2)).Select

    Selection.Font.Color = RGB(TortueR, TortueG, TortueB)

    Selection.Font.Name = Police ' "Chicago"

    Selection.Font.Size = taille

    Selection.Font.Bold = Gras

    Selection.Font.Italic = Italique

    Selection.TypeText Text:=chaine

    Selection.ShapeRange.Line.Visible = msoFalse

    Selection.ShapeRange.Fill.Visible = msoFalse

End Sub

Sub Etiquette()

    ActiveDocument.Shapes.AddTextbox(msoTextOrientationHorizontal, 330, 110, 20, 20).Select

    Selection.TypeText Text:="?"

    Selection.ShapeRange.Line.Visible = msoFalse

    Selection.ShapeRange.Fill.Visible = msoFalse

End Sub

Sub RectangleGen(x1, y1, x2, y2, Remplir)

    If Remplir Then DebutRemplir

    Segment x1, y1, x2, y1

    Segment x2, y1, x2, y2

    Segment x2, y2, x1, y2

    Segment x1, y2, x1, y1

    If Remplir Then FinRemplir

End Sub

Sub rectangle(x1, y1, x2, y2)

    RectangleGen x1, y1, x2, y2, False

End Sub

Sub RectanglePlein(x1, y1, x2, y2)

    RectangleGen x1, y1, x2, y2, True

End Sub

Function Hasard(a, b)

    Hasard = Int(a + (b + 1 - a) * Rnd())

End Function

Function CosD(x)

    CosD = Cos(x * pi / 180)

End Function

Function SinD(x)

    SinD = Sin(x * pi / 180)

End Function

Function TanD(x)

    TanD = Tan(x * pi / 180)

End Function

Function unitesAcm(x)

    unitesAcm = x * (2.54 / 72)

End Function

Function cmAunites(x)

    cmAunites = x * (72 / 2.54)

End Function

Sub Grille(x, y, dx, dy, nx, ny)

    Dim i As Integer

    For i = 0 To nx

        Segment x + i * dx, y, x + i * dx, y + ny * dy

    Next i

    For i = 0 To ny

        Segment x, y + i * dy, x + nx * dx, y + i * dy

    Next i

End Sub

Sub ProPoint(x, y, r, G, b, E)

    CC r / 256, G / 256, b / 256

    CR r / 256, G / 256, b / 256

    TC 1

    disque ProFacteur * x, -ProFacteur * y, ProFacteur * E / 2

End Sub

Sub ProSegment(x1, y1, x2, y2, r, G, b, E)

    CC r / 256, G / 256, b / 256

    TC ProFacteur * E

    Segment ProFacteur * x1, -ProFacteur * y1, ProFacteur * x2, -ProFacteur * y2

End Sub

Sub ProCercle(x, y, rayon, r, G, b, E)

    CC r / 256, G / 256, b / 256

    TC ProFacteur * E

    Cercle ProFacteur * x, -ProFacteur * y, ProFacteur * rayon

End Sub

Sub ProArc(x, y, rayon, debut, fin, r, G, b, E)

    CC r / 256, G / 256, b / 256

    TC ProFacteur * E

    Arc ProFacteur * x, -ProFacteur * y, ProFacteur * rayon, 360 - fin, 360 - debut

End Sub

Sub ProClip(xmin, xmax, ymin, ymax)

    CR 255, 255, 200

    RectangleGen ProFacteur * xmin, -ProFacteur * ymin, ProFacteur * xmax, -ProFacteur * ymax, True

End Sub

Sub InitFonctions()

    FonctionsPas = 1

    FonctionsPointsRelies = True

    FonctionsXmin = -10

    FonctionsXmax = 10

    FonctionsYmin = -10

    FonctionsYmax = 10

    FenetreFonctions -50, 50, -25, 25

End Sub

Sub RegionFonctions(xmin, xmax, ymin, ymax)

    FonctionsXmin = xmin

    FonctionsXmax = xmax

    FonctionsYmin = ymin

    FonctionsYmax = ymax

    CalculCoefficientsFonctions

End Sub

Sub FenetreFonctions(gauche_, droite_, bas_, haut_)

    FenetreXmin = gauche_

    FenetreXmax = droite_

    FenetreYmin = bas_

    FenetreYmax = haut_

    CalculCoefficientsFonctions

End Sub

Sub DimGraphesFonctions(base, hauteur)

    FenetreFonctions -base / 2, base / 2, -hauteur / 2, hauteur / 2

End Sub

Sub GrilleFonctions(nx, ny)

    Dim i As Integer, d As Double

    If (nx < 1) Or (ny < 1) Then Exit Sub

    d = (FenetreXmax - FenetreXmin) / nx

    For i = 0 To nx

        Segment FenetreXmin + i * d, FenetreYmin, FenetreXmin + i * d, FenetreYmax

    Next i

    d = (FenetreYmax - FenetreYmin) / ny

    For i = 0 To ny

        Segment FenetreXmin, FenetreYmin + i * d, FenetreXmax, FenetreYmin + i * d

    Next i

End Sub

Sub CalculCoefficientsFonctions()

'x = FonctionsMX * xE + FonctionsBX  ;  y = FonctionsMY * xE + FonctionsBY  ;  xE = FonctionsMXE * x + FonctionsBXE  ;  yE = FonctionsMYE * x + FonctionsBYE

    FonctionsMX = (FonctionsXmax - FonctionsXmin) / (FenetreXmax - FenetreXmin)

    FonctionsBX = FonctionsXmin - FenetreXmin * FonctionsMX

    FonctionsMXE = 1 / FonctionsMX

    FonctionsBXE = FenetreXmin - FonctionsXmin * FonctionsMXE

    FonctionsMYE = (FenetreYmax - FenetreYmin) / (FonctionsYmax - FonctionsYmin)

    FonctionsBYE = FenetreYmin - FonctionsYmin * FonctionsMYE

    FonctionsMY = 1 / FonctionsMYE

    FonctionsBY = FonctionsYmin - FenetreYmin * FonctionsMY

End Sub

Sub TracerAxes()

    yE = ConvYE(0)

    If FenetreYmin <= yE And yE <= FenetreYmax Then

        Segment FenetreXmin, yE, FenetreXmax, yE

        Selection.ShapeRange.Line.BeginArrowheadStyle = msoArrowheadNone

        Selection.ShapeRange.Line.EndArrowheadStyle = msoArrowheadTriangle

        Selection.ShapeRange.Line.EndArrowheadWidth = msoArrowheadWidthMedium

        Selection.ShapeRange.Line.EndArrowheadLength = msoArrowheadLengthMedium

        Selection.ShapeRange.Line.Visible = msoTrue

        xE = ConvXE(0)

    End If

    If FenetreXmin <= xE And xE <= FenetreXmax Then

        Segment xE, FenetreYmin, xE, FenetreYmax

        Selection.ShapeRange.Line.BeginArrowheadStyle = msoArrowheadNone

        Selection.ShapeRange.Line.EndArrowheadStyle = msoArrowheadTriangle

        Selection.ShapeRange.Line.EndArrowheadWidth = msoArrowheadWidthMedium

        Selection.ShapeRange.Line.EndArrowheadLength = msoArrowheadLengthMedium

    End If

End Sub

Function ConvXE(x)

    ConvXE = FonctionsMXE * x + FonctionsBXE

End Function

Function ConvYE(y)

    ConvYE = FonctionsMYE * y + FonctionsBYE

End Function

Sub PasFonctions(a)

    FonctionsPas = a

End Sub

Sub PointsRelies()

    FonctionsPointsRelies = True

End Sub

Sub PointsNonRelies()

    FonctionsPointsRelies = False

End Sub

Sub TracerFonctionAux(NumeroFonction, Pas, Lier)

    Dim yE0 As Variant, xE As Double, x As Double, y As Variant, yE As Double

    yE0 = "ND"

    For xE = FenetreXmin To FenetreXmax Step Pas

        x = FonctionsMX * xE + FonctionsBX

        y = F(NumeroFonction, x)

        If y = "ND" Then  ' ND = Non Défini

            yE0 = "ND"

        Else

            If Not Lier Then

                If (FonctionsYmin <= y) And (y <= FonctionsYmax) Then

                    yE = FonctionsMYE * y + FonctionsBYE

                    disque xE, yE, TortueEpais / 2

                End If

            Else

                If y < FonctionsYmin Then y = FonctionsYmin

                If y > FonctionsYmax Then y = FonctionsYmax

                yE = FonctionsMYE * y + FonctionsBYE

                Select Case yE0

                    Case "ND"

                        Segment xE, yE, xE, yE

                    Case Else

                        Segment xE - Pas, yE0, xE, yE

                End Select

            End If

            yE0 = yE

        End If

    Next xE

    CadreFonctions

End Sub

Sub TracerFonction(NumeroFonction)

    If ModeDemo Then

        TracerFonctionAuxDemo NumeroFonction, FonctionsPas, FonctionsPointsRelies

    Else

        TracerFonctionAux NumeroFonction, FonctionsPas, FonctionsPointsRelies

    End If

End Sub

Sub CadreFonctions()

    Dim r As Integer, G As Integer, b As Integer, E As Integer

    r = TortueR

    G = TortueG

    b = TortueB

    E = TortueEpais

    TortueR = 0

    TortueG = 0

    TortueB = 0

    ' TortueEpais = 1

    rectangle FenetreXmin, FenetreYmax, FenetreXmax, FenetreYmin

    TortueR = r

    TortueG = G

    TortueB = b

    TortueEpais = E

End Sub

Sub FaireFilm(nb, x0, y0, x1, y1, r, v, b)

    NbTotalImagesFilm = nb

    InitFilm

    If r = 255 And v = 255 And b = 255 Then r = 254

    For noImage = 1 To NbTotalImagesFilm

        CR r, v, b

        RectanglePlein x0, y0, x1, y1

        If ModeDemo Then

            TracerImageDemo noImage

        Else

            TracerImage noImage

        End If

        AjouterImageAuFilm

    Next noImage

    FermerSansSauver

End Sub

Sub InitFilm()

    NomFenetre = ActiveDocument.Windows(1).Caption

    NouveauDoc

    NomFenetreFilm = ActiveDocument.Windows(1).Caption

    For i = 1 To NbTotalImagesFilm - 1

        Selection.InsertBreak Type:=wdPageBreak

    Next i

    Selection.MoveUp Unit:=wdLine, Count:=NbTotalImagesFilm

    Windows(NomFenetre).Activate

End Sub

Sub AjouterImageAuFilm()

    ActiveDocument.Shapes.SelectAll

    On Error Resume Next

    Selection.ShapeRange.Group.Select

    On Error GoTo 0

    Selection.Copy

    Windows(NomFenetreFilm).Activate

    Selection.Paste

    Selection.Collapse

    Selection.MoveDown Unit:=wdLine, Count:=1

    Windows(NomFenetre).Activate

    VE

End Sub

Sub NouveauDoc()

    Documents.Add DocumentType:=wdNewBlankDocument

    Windows(ActiveDocument.Windows(1).Caption).Activate

End Sub

Sub FermerSansSauver()

    ActiveWindow.Close SaveChanges:=wdDoNotSaveChanges

End Sub

Sub Figure() ' Appeler (mais ne pas modifier) cette procédure pour tracer

    Init

    NouveauDoc 'VE

    InitFonctions

    Application.ScreenUpdating = MiseAuPointLG

    NbTotalImagesFilm = 0

    ModeDemo = False

    Graphique 'Application.Run MacroName:="Graphique"

    If NbTotalImagesFilm = 0 Then

        ActiveDocument.Shapes.SelectAll

        On Error Resume Next

        Selection.ShapeRange.Group.Select

        On Error GoTo 0

    End If

End Sub

'%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%%

Sub Demo()

    Dim i As Integer

    ActiveWindow.View.Type = wdPageView

    ActiveWindow.ActivePane.View.Zoom.PageFit = wdPageFitFullPage

    ModeDemo = True

        nb = InputBox("Nombre de triangles:" + Chr(13) + " (Entre 6 et 30)", "Spirale de Pythagore")

        If Not IsNumeric(nb) Then Exit Sub

        nb = recadrer(CInt(nb), 1, 30)

        SpiralePythagore 70, nb

    If ArretP Then Exit Sub

        nb = InputBox("Nombre de perles:" + Chr(13) + " (Entre 2 et 100)", "Colliers")

        If Not IsNumeric(nb) Then Exit Sub

        nb = recadrer(CInt(nb), 2, 150)

        Collier 250, nb

    If ArretP Then Exit Sub

        nb = InputBox("Nombre de demi-cercles:" + Chr(13) + " (Entre 1 et 50)", _

            "Approcher segment par demi-cercles")

        If Not IsNumeric(nb) Then Exit Sub

        nb = recadrer(CInt(nb), 1, 100)

        Approx 275, nb

    If ArretP Then Exit Sub

        nb = InputBox("Nombre max de niveaux:" + Chr(13) + " (Entre 0 et 5)", "Flocon de Koch")

        If Not IsNumeric(nb) Then Exit Sub

        nb = recadrer(CInt(nb), 0, 5)

        FigureFlocon nb

    If ArretP Then Exit Sub

        nb = InputBox("Nombre max de niveaux:" + Chr(13) + " (Entre 0 et 5)", "Triangle de Sierpinski")

        If Not IsNumeric(nb) Then Exit Sub

        nb = recadrer(CInt(nb), 0, 5)

        For i = 1 To nb + 1

            VE

            LC

            FPos -300, -200

            BC

            Droite 30

            SierpinskiRemplis 600, i

            ActiveDocument.Shapes.SelectAll

            On Error Resume Next

            Selection.ShapeRange.Group.Select

            On Error GoTo 0

            Application.ScreenUpdating = True

            Application.ScreenRefresh

            If i <= nb Then Rep = MsgBox("Niveau suivant?", vbOKOnly, "Niveau " + CStr(i - 1))

        Next i

    If ArretP Then Exit Sub

        nb = InputBox("Nombre d'approximations:" + Chr(13) + " (Entre 0 et 4)" + Chr(13), _

        "Approcher sinus par des polynômes")

        If Not IsNumeric(nb) Then Exit Sub

        nb = recadrer(CInt(nb), 0, 4)

        Fonctions (nb)

    If ArretP Then Exit Sub

        nb = InputBox("Nombre max de colliers:" + Chr(13) + " (Entre 1 et 30)", "Colliers (film)")

        If Not IsNumeric(nb) Then Exit Sub

        nb = recadrer(CInt(nb), 1, 50)

        FaireFilm nb, -100, -100, 100, 100, 255, 255, 0

        MsgBox "Il suffit d'enregistrer en tant que page web.", vbOKOnly, "Pour obtenir une suite d'images..."

End Sub

Function ArretP()

    Application.ScreenUpdating = True

    Application.ScreenRefresh

    Rep = MsgBox("On continue?", vbYesNo, "Message")

    ArretP = (Rep = vbNo)

    If Not ArretP Then

        FermerSansSauver

        NouveauDoc

        ActiveWindow.ActivePane.View.Zoom.PageFit = wdPageFitFullPage

        Init

    End If

End Function

Sub DemoNo(n)

    ModeDemo = True

    ActiveWindow.View.Type = wdPageView

    ActiveWindow.ActivePane.View.Zoom.PageFit = wdPageFitFullPage

    Select Case n

        Case 0

            MenuDemo

        Case 1

            nb = InputBox("Nombre de triangles:" + Chr(13) + " (Entre 6 et 30)", "Spirale de Pythagore")

            If Not IsNumeric(nb) Then Exit Sub

            nb = recadrer(CInt(nb), 1, 30)

            SpiralePythagore 70, nb

        Case 2

            nb = InputBox("Nombre de perles:" + Chr(13) + " (Entre 2 et 100)", "Colliers")

            If Not IsNumeric(nb) Then Exit Sub

            nb = recadrer(CInt(nb), 2, 150)

            Collier 250, nb

        Case 3

            nb = InputBox("Nombre de demi-cercles:" + Chr(13) + " (Entre 1 et 50)", _

            "Approcher segment par demi-cercles")

            If Not IsNumeric(nb) Then Exit Sub

            nb = recadrer(CInt(nb), 1, 100)

            Approx 275, nb

        Case 4

            nb = InputBox("Nombre de niveaux:" + Chr(13) + " (Entre 0 et 5)", "Flocon de Koch")

            If Not IsNumeric(nb) Then Exit Sub

            nb = recadrer(CInt(nb), 0, 5)

            FigureFlocon nb

        Case 5

            nb = InputBox("Nombre de niveaux:" + Chr(13) + " (Entre 0 et 5)", "Triangle de Sierpinski")

            If Not IsNumeric(nb) Then Exit Sub

            nb = recadrer(CInt(nb), 0, 5)

            LC

            FPos -300, -200

            BC

            Droite 30

            SierpinskiRemplis 600, nb + 1

        Case 6

            nb = InputBox("Nombre d'approximations:" + Chr(13) + " (Entre 0 et 4)" + Chr(13), _

            "Approcher sinus par des polynômes")

            If Not IsNumeric(nb) Then Exit Sub

            nb = recadrer(CInt(nb), 0, 4)

            Fonctions (nb)

        Case 7

            nb = InputBox("Nombre max de colliers:" + Chr(13) + " (Entre 1 et 30)", "Colliers (film)")

            If Not IsNumeric(nb) Then Exit Sub

            nb = recadrer(CInt(nb), 1, 50)

            FaireFilm nb, -100, -100, 100, 100, 255, 255, 0

        Case Else

    End Select

End Sub

Function recadrer(x, a, b)

    recadrer = x

    If x < a Then recadrer = a

    If x > b Then recadrer = b

End Function

Sub MenuDemo() '

    Selection.Font.Size = 18

    Selection.Font.Bold = True

    Selection.TypeText Text:="Liste des démonstrations"

    Selection.TypeParagraph

    Selection.Font.Size = 12

    Selection.Font.Bold = False

    Selection.TypeText Text:=" • DemoNo 0: Afficher le présent menu."

    Selection.TypeParagraph

    Selection.TypeText Text:=" • DemoNo 1: Spirale de Pythagore."

    Selection.TypeParagraph

    Selection.TypeText Text:=" • DemoNo 2: Colliers."

    Selection.TypeParagraph

    Selection.TypeText Text:=" • DemoNo 3: Approcher un segment par des demi-cercles."

    Selection.TypeParagraph

    Selection.TypeText Text:=" • DemoNo 4: Flocon de Koch."

    Selection.TypeParagraph

    Selection.TypeText Text:=" • DemoNo 5: Triangle de Sierpinski."

    Selection.TypeParagraph

    Selection.TypeText Text:=" • DemoNo 6: Approcher sinus par des polynômes."

    Selection.TypeParagraph

    Selection.TypeText Text:=" • DemoNo 7: Suite d'images à réunir dans un film (ou un GIF animé)."

End Sub

Sub SpiralePythagore(cote, nombre)

    Dim x As Double, y As Double, Angle As Double, i As Integer

    x = cote

    y = 0

    Angle = 0

    For i = 1 To nombre

        CC 256 * Rnd(), 256 * Rnd(), 256 * Rnd()

        CR 256 * Rnd(), 256 * Rnd(), 256 * Rnd()

        DebutRemplir

        FPos x, y

        Dr -Angle

        Av cote

        x = TortueX

        y = TortueY

        FPos 0, 0

        FinRemplir

        Dessous

        Angle = Atn(1 / Sqr(i + 1)) * 180 / 3.14159265

    Next i

End Sub

Sub Collier(r, n)

    'Cercle 0, 0, R

    r1 = r * Sin(3.14159265 / n) / (1 + Sin(3.14159265 / n))

    For i = 0 To n - 1

        CR Hasard(0, 255), Hasard(0, 255), Hasard(0, 255)

        disque (r - r1) * Cos(i * 2 * pi / n), (r - r1) * Sin(i * 2 * 3.14159265 / n), r1

    Next i

    'Cercle 0, 0, R - 2 * R1

End Sub

Sub Approx(xmax, n)

    Dim i As Integer

    Segment -xmax, 0, xmax, 0

    r = xmax / (1 * n)

    x0 = -xmax + r

    For i = 1 To n

        If (i Mod 2) = 1 Then Arc x0, 0, r, 0, 180 Else Arc x0, 0, r, 180, 0

        x0 = x0 + 2 * r

    Next i

End Sub

Sub ExApprox(xmax, n)

    Dim i As Integer

    Segment 0, 0, xmax, 0

    r = xmax / (2 * n)

    x0 = r

    For i = 1 To n

        If (i Mod 2) = 1 Then Arc x0, 0, r, 0, 180 Else Arc x0, 0, r, 180, 0

        x0 = x0 + 2 * r

    Next i

End Sub

Sub FigureFlocon(niv)

    Dim i As Integer

    For i = 0 To niv

        'Application.ScreenUpdating = False ' Mettre en commentaire si on veut voir tracer pas à pas

        VE

        LC

        FPos -300, -200

        BC

        CR 255 * Rnd(), 255 * Rnd(), 255 * Rnd()

        DebutRemplir ' INVESTIGUER: remplir accélère les tracés

        DessinFlocon 600, i

        FinRemplir

        ActiveDocument.Shapes.SelectAll

        On Error Resume Next

        Selection.ShapeRange.Group.Select

        On Error GoTo 0

        Application.ScreenUpdating = True

        Application.ScreenRefresh

        If i < niv Then Rep = MsgBox("Flocon suivant?", vbOKOnly, "Niveau " + CStr(i))

    Next i

End Sub

Sub DessinFlocon(taille, n)

    Dim i As Integer

    Dr 30

    For i = 1 To 3

        Flocon taille, n

        Dr 120

    Next i

End Sub

Sub Flocon(taille, NbNiveaux)

    If NbNiveaux = 0 Then

        Av taille

        Exit Sub

    End If

    Flocon taille / 3, NbNiveaux - 1

    Dr -60

    Flocon taille / 3, NbNiveaux - 1

    Dr 120

    Flocon taille / 3, NbNiveaux - 1

    Dr -60

    Flocon taille / 3, NbNiveaux - 1

End Sub

Sub SierpinskiRemplis(d, E)

    Dim i As Integer

    If E = 0 Then Exit Sub

    If E = 1 Then

        BC

        DebutRemplir

        CR 255 * Rnd(), 255 * Rnd(), 255 * Rnd()

    End If

    For i = 1 To 3

        SierpinskiRemplis (d / 2), E - 1

        Av d

        Dr 120

    Next i

    If E = 1 Then

        FinRemplir

        LC

    End If

End Sub

Sub Fonctions(n)

    RegionFonctions -pi, pi, -2, 2

    FenetreFonctions -200, 200, -150, 150

    TracerAxes

    PasFonctions 5

    TC 2

    CC 0, 0, 0

    TracerFonction 1

    For k = 2 To n + 1

        Select Case k

            Case 2

                CC 0, 0, 255

            Case 3

                CC 0, 255, 0

            Case 4

                CC 0, 255, 255

            Case 5

                CC 255, 0, 0

            Case Else

        End Select

        TracerFonction k

    Next k

End Sub

Sub TracerFonctionAuxDemo(NumeroFonction, Pas, Lier)

    Dim yE0 As Variant, xE As Double, x As Double, y As Variant, yE As Double

    yE0 = "ND"

    For xE = FenetreXmin To FenetreXmax Step Pas

        x = FonctionsMX * xE + FonctionsBX

        y = Fdemo(NumeroFonction, x)

        If y = "ND" Then  ' ND = Non Défini

            yE0 = "ND"

        Else

            If y < FonctionsYmin Then y = FonctionsYmin

            If y > FonctionsYmax Then y = FonctionsYmax

            yE = FonctionsMYE * y + FonctionsBYE

            If Not Lier Then

                Segment xE, yE, xE, yE 'Disque Xecran, Yecran, Taille

            Else

                Select Case yE0

                    Case "ND"

                        Segment xE, yE, xE, yE

                    Case Else

                        Segment xE - Pas, yE0, xE, yE

                End Select

            End If

            yE0 = yE

        End If

    Next xE

    CadreFonctions

End Sub

Function Fdemo(n, x)

    Fdemo = "ND"

    On Error GoTo Erreur

    Select Case n

        Case 1

            Fdemo = Sin(x)

        Case 2

            Fdemo = x

        Case 3

            Fdemo = x - x * x * x / 6

        Case 4

            Fdemo = x - x * x * x / 6 + x * x * x * x * x / 120

        Case 5

            Fdemo = x - (x ^ 3) / fac(3) + (x ^ 5) / fac(5) - (x ^ 7) / fac(7)

        Case 6

        Case 7

        Case 8

        Case 9

        Case 10

        Case Else

    End Select

Erreur:

End Function

Function fac(n)

    Dim k As Integer, p As Integer

    p = 1

    For k = 2 To n

        p = p * k

    Next k

    fac = p

End Function

Sub TracerImageDemo(n)

    Collier 80, n + 3

End Sub

Sub TracerImageDemoAlt(n) ' n entre 1 et 6

    LC

    FPos -95, -85

    BC

    Droite 30

    SierpinskiRemplis 190, n

End Sub

Sub TracerFonctionIntervalle(NumeroFonction, abscisseMin, abscisseMax)

    Dim yE0 As Variant, xE As Double, x As Double, y As Variant, yE As Double

    yE0 = "ND"

    Pas = FonctionsPas

    Lier = FonctionsPointsRelies

    intervallexmin = Round((abscisseMin - FonctionsBX) / FonctionsMX)

    intervallexmin = max(FenetreXmin, min(FenetreXmax, intervallexmin))

    intervallexmax = Round((abscisseMax - FonctionsBX) / FonctionsMX)

    intervallexmax = max(FenetreXmin, min(FenetreXmax, intervallexmax))

    For xE = intervallexmin To intervallexmax Step Pas

        x = FonctionsMX * xE + FonctionsBX

        y = F(NumeroFonction, x)

        If y = "ND" Then  ' ND = Non Défini

            yE0 = "ND"

        Else

            If Not Lier Then

                If (FonctionsYmin <= y) And (y <= FonctionsYmax) Then

                    yE = FonctionsMYE * y + FonctionsBYE

                    disque xE, yE, TortueEpais / 2

                End If

            Else

                If y < FonctionsYmin Then y = FonctionsYmin

                If y > FonctionsYmax Then y = FonctionsYmax

                yE = FonctionsMYE * y + FonctionsBYE

                Select Case yE0

                    Case "ND"

                        Segment xE, yE, xE, yE

                    Case Else

                        Segment xE - Pas, yE0, xE, yE

                End Select

            End If

            yE0 = yE

        End If

    Next xE

    CadreFonctions

End Sub

Function min(x, y)

    If x < y Then min = x Else min = y

End Function

Function max(x, y)

    If x > y Then max = x Else max = y

End Function

Function Round(x)

    Round = Int(x + 0.5)

End Function

